
Stat 342  Lab Week 7 - SAS/IML 

 

This lab is on the IML (Interactive Matrix Language) system that is accessed through the procedure, 

'proc iml'. This lab does not require any external datasets. 

 

Textbook source: Pages 62-68 of 'SAS and R'. 

Additional source for interest:  

SAS Whitepaper 144-2013 - Getting started with the SAS/IML Language, by Rick Wilkin 

https://support.sas.com/resources/papers/proceedings13/144-2013.pdf 

 

Part I, Defining matrices 

 

1. Define and print 
 

The IML procedure is like a data step in that you can alter datasets or make them from scratch in it. 

However, The 'base unit' of a proc iml isn't a single value, it's a matrix (which could be a 1x1 matrix if 

you really wanted). 

 

 

To demonstrate this, use the IML procedure below to create a 5x1 height matrix. 
 

 

proc iml;  

 height = {14,10,8,6,4}; /* CURLY BRACES */ 

 print height; 
quit; /* NOT 'RUN' */ 

 

 

 

Now try creating multiple 5x1 matrices and printing them together 
 

proc iml;  

 height = {14,10,8,6,4}; 

 width = {8,2,-3,.,6}; 

 depth = {4,3,5,2,.}; 

 volume = height # width # depth; 

 print depth width height volume; /* print order matters */  

quit;  

  

 

 

 

 

 

  



2. The use of commas 
  

When defining a matrix, a comma marks the end of a row. Try the box dimensions code withspaces 

instead of commas between entries in the matrix definitions. This creates 1x5 matrices instead of 5x1 

ones, but the print command still tries to concatenate them together horizontally. 
 

proc iml;  

 height = {14 10 8 6 4}; 

 width = {8 2 -3 . 6}; 

 depth = {4 3 5 2 .}; 

 volume = height # width # depth; 

 print depth width height volume;  

quit;  

  
 

When listing multiple matrices in a print command, if there is ONLY space between matrices, print 

will concatenate them together as well as it can. If there is a comma between them, it will output those 

(lists of) matrices separately. 
 

proc iml;  

 height = {14 10 8 6 4}; 

 width = {8 2 -3 . 6}; 

 depth = {4 3 5 2 .}; 

 volume = height # width # depth; 

 print depth, width, height, volume;  

quit;  

  
 

3. Exploration of other cases 
 

What happens when some but not all of the spaces between values / names are commas?  

 

Modify at least one version of one of the programs in task 2 above to see what happens when some 

values in a matrix definition are separated by commas, but not all. 

 

Do the same for one of the print commands. 

 

 

What happens when not all the matrices are the same dimensions? Add or remove some elements from 

one of the matrices in one of the programs above and run it. What happens? 

 

 

 

 

 

 

 

 

 



4. Formatting 
  

You can add formatting to a matrix being output with some extra settings in square braces [] at the end 

of the print command. 
 

proc iml;  

 height = {14,10,8,6,4}; 

 width = {8,2,-3,.,6}; 

 depth = {4,3,5,2,.}; 

 volume = height # width # depth; 

 print depth width height volume  

  [rowname = {"Box A", "Box B", "Box C", "Box D", "Box E"} 

   colname = {"H" "W" "D" "Vol"} 

   label = "Jinko's shipping and copying"];  

quit;  

  
 

 

5. The shape function, and the : range definer 
  

You don't need to specify the dimensions or row-endings of a matrix in advance. You can change them 

on the fly with the shape() function. 

 

Define a 20x1 matrix with 1:12, which is IML shorthand for 1,2,3, ... , 10,11,12 (just like in R), and see 

what happens when you use the shape function on that matrix.  

 

 
proc iml;  

 x = 1:12; 

 onerow = shape(x,1); /* (input, Nrows) */ 

 m1 = shape(x,2);  

 m2 = shape(x,2,6); /* (input, Nrows, Ncols) */ 

 m3 = shape(x,4,3); 

 print x,onerow,m1,m2,m3; 

quit;  

  
 

  



6. Matrix operations, element-wise 
  

The symbols +, -, #, and / do addition, subtract, multiplication, and division between the elements in 

two matrices of the same dimensions. These are element-wise operations, meaning operation is 

performed between the first element of each matrix to get the first element of the result matrix and so 

on. Different elements in a matrix are completely independent. 

 

 
proc iml;  

 x = {-4 9 , 2 5 , 8 7}; 

 y = {1 0  , . 4 , 3 -1}; 

 elem_sum = x + y; 

 elem_dif = 2#x - y; 

 elem_product = x#y; 

 elem_divis = x/y; 

 print x, y, elem_sum, elem_dif, elem_product, elem_divis; 

quit;  

  
 

 

 

7. Matrix operations, matrix-wise 
  

The symbol * performs the classical matrix multiplication on two compatible matrices. Each element of 

the result is an inner product of the row of the first matrix and the column of the second matrix. 

 

The t() function transposes a matrix, which is akin to flipping it on its side (bottom-left to top-right). 

 

The inv() function can be used to find the inverse of a square matrix, if one exists. 

 

For more information on matrix multiplication, inversion, and transposition, consult a linear algebra or 

matrix algebra textbook. 

 

 
proc iml;  

 A = {1 2 , 3 4}; 

 B = {0 1 , 2 3}; 

 Asq = A * A; 

 AB = A * B; 

 BA = B * A; 

 At = t(A); 

 AtB = t(A) * B; 

 

 Ainv = inv(A); 

 identity = Ainv * A; 

 

 print A,B,Asq,AB,BA,At,AtB,Ainv,identity; 

quit;  

  


